

# TIPS Y TRUCOS DE EXCEL

Consejos que te convertirán en un experto de Excel

# TIPS Y TRUCOS DE EXCEL

---

Por Moisés Ortíz

## TABLA DE CONTENIDO

1. FÓRMULA PARA BUSCAR DATOS .....	1
2. BUSCAR VALORES USANDO FORMATO CONDICIONAL.....	7
3. CÓMO COMPARAR DOS LISTAS EN EXCEL .....	11
4. FÓRMULA PARA OBTENER EL ÚLTIMO DÍA DEL MES.....	16
5. FÓRMULA PARA OBTENER EL DÍA DE LA SEMANA.....	18
6. FÓRMULA PARA CALCULAR LA EDAD.....	24
7. LA FUNCIÓN SIFECHA EN EXCEL .....	26
8. SUMAR RANGOS QUE CONTIENEN ERRORES.....	29
9. FÓRMULA PARA CONTAR PALABRAS EN EXCEL .....	33
10. LISTA DESPLEGABLE CON DATOS DE OTRA HOJA .....	37
11. LISTAS DESPLEGABLES DEPENDIENTES .....	43
12. CARGAR VALORES DE LISTA DINÁMICAMENTE.....	54

## 1. FÓRMULA PARA BUSCAR DATOS

Una de las acciones más comunes en Excel es la de buscar un valor dentro de una columna, razón por la cual la función BUSCARV es una de las funciones más utilizadas, sin embargo no es la única alternativa que tenemos para buscar en Excel.

Para aquellos que apenas inician en Excel haré un breve recordatorio de la [función BUSCARV](#) la cual tiene tres argumentos obligatorios y uno opcional.

	A	B	C	D	E	F	G
1	Nombre:	Diana		Nombre	Apellido	Teléfono	
2	Teléfono:	982-3433		Antonio	Zamora	473-7110	
3				Bernardo	Jaramillo	652-8968	
4				Diana	Cardona	982-3433	
5				Enrique	Montes	901-9724	
6				Celia	Toledo	288-8167	
7							

1. El primer argumento de la función BUSCARV indica el valor que vamos a buscar. En el ejemplo nos referimos al valor de la celda B1 que es “Diana”.
2. El segundo argumento es el rango donde se realizará la búsqueda y el cual debe cumplir con dos condiciones. La primera condición es que la primera columna del rango debe ser la columna donde se realizará la búsqueda. La segunda condición es que la columna con el valor que necesitamos como resultado debe estar contenida en dicho rango. En nuestro ejemplo, deseamos encontrar el teléfono de “Diana” y por lo tanto la columna F debe estar incluida en el rango de búsqueda.

3. El tercer argumento indica la columna que deseamos obtener como resultado. Nuestro rango (D2:F6) contiene tres columnas, la primera es el Nombre, la segunda el Apellido y la tercera el Teléfono. Ya que necesitamos como resultado la columna Teléfono, es necesario indicar el valor 3 como el tercer argumento de la función.
4. El cuarto argumento de la función BUSCARV es opcional e indica el tipo de búsqueda que deseamos realizar. En nuestro ejemplo utilizamos el valor FALSO para indicar a la función que deseamos una coincidencia exacta del valor buscado.

Como resultado, la función BUSCARV nos devuelve el valor de la columna Teléfono que corresponde a “Diana”. Con solo cambiar el valor en la celda B1 obtendremos el teléfono de la persona indicada:

B2		=BUSCARV(B1, D2:F6, 3, FALSO)					
	A	B	C	D	E	F	G
1	Nombre:	Bernardo		Nombre	Apellido	Teléfono	
2	Teléfono:	652-8968		Antonio	Zamora	473-7110	
3				Bernardo	Jaramillo	652-8968	
4				Diana	Cardona	982-3433	
5				Enrique	Montes	901-9724	
6				Celia	Toledo	288-8167	
7							

Considerando el mismo ejemplo anterior, si deseo encontrar el teléfono de una persona basado en su apellido, no puedo hacerlo utilizando la misma fórmula ya que la función BUSCARV siempre realizará la búsqueda en la primera columna. Para hacer este tipo de búsqueda debo modificar la fórmula de la siguiente manera:

B2		fx =BUSCARV(B1, E2:F6, 2, FALSO)					
	A	B	C	D	E	F	G
1	Apellido:	Zamora		Nombre	Apellido	Teléfono	
2	Teléfono:	473-7110		Antonio	Zamora	473-7110	
3				Bernardo	Jaramillo	652-8968	
4				Diana	Cardona	982-3433	
5				Enrique	Montes	901-9724	
6				Celia	Toledo	288-8167	
7							

Los dos cambios realizados a la fórmula son los siguientes:

- El segundo argumento cambió a E2:F6 porque la función BUSCARV siempre busca en la primera columna, y ya que deseo buscar por apellido, la columna E debe ser la primera en el rango de búsqueda.
- Debido a la modificación anterior es necesario cambiar también el tercer argumento de la función ya que el rango de búsqueda se redujo a la columna Apellido y a la columna Teléfono. Ahora nuestro resultado deseado se encuentra en la segunda columna del rango de búsqueda y por lo tanto debo indicar el valor numérico 2 en el tercer argumento.

El hecho de que la función BUSCARV solo nos permita buscar en la primera columna de un rango representa una desventaja para resolver ciertos casos. Por ejemplo, considera el escenario donde deseamos buscar en nuestros datos de ejemplo el nombre de una persona a partir de su apellido. Esto implica buscar un valor en la columna E y devolver el valor correspondiente de la columna D lo cual es imposible de resolver con la función BUSCARV a menos de que intercambiamos el orden de las columnas moviendo la columna Apellido a la izquierda de la columna Nombre.

En ocasiones no es posible realizar estos cambios en el orden de las columnas y por ello es importante conocer y tener presente un método

alternativo que nos permitirá crear una fórmula para buscar datos en Excel sin hacer uso de la función BUSCARV.

Antes de continuar debemos comprender el funcionamiento de la [función COINCIDIR](#) la cual busca un valor y nos devuelve su posición dentro del rango de búsqueda. Esta función tiene tres argumentos y solo dos de ellos son obligatorios.

	A	B	C	D	E	F	G
1	Nombre:	Diana		Nombre	Apellido	Teléfono	
2	Posición:	3		Antonio	Zamora	473-7110	
3				Bernardo	Jaramillo	652-8968	
4				Diana	Cardona	982-3433	
5				Enrique	Montes	901-9724	
6				Celia	Toledo	288-8167	
7							

1. El primer argumento de la función COINCIDIR es el valor a buscar. En el ejemplo anterior será el valor de la celda B1.
2. El segundo argumento es el rango de búsqueda y que para nuestro ejemplo son las celdas que contienen todos los nombres.
3. El tercer argumento es opciones y nos permite especificar el tipo de búsqueda a realizar. Para obtener una coincidencia exacta debemos especificar el valor cero.

Como resultado obtenemos el valor 3 que nos indica que “Diana” es el tercer nombre dentro del rango D2:D6. El hecho de obtener esta posición es de mucha ayuda porque si queremos conocer el teléfono de “Diana” debemos obtener el valor en la posición 3 bajo la columna Teléfono y eso es precisamente lo que haremos en el siguiente paso.

Ahora que ya sabemos cómo encontrar la posición de un valor con la función COINCIDIR, solo nos falta encontrar el valor correspondiente en otra columna y para eso utilizamos la [función INDICE](#). La forma más simple de esta función requiere de solo dos argumentos, el primero de ellos es el rango de búsqueda y el segundo será la posición, dentro de dicho rango, que deseamos obtener.

B2      fx      =INDICE(F2:F6,COINCIDIR(B1,D2:D6,0))							
	A	B	C	D	E	F	G
1	Nombre:	Diana		Nombre	Apellido	Teléfono	
2	Teléfono:	982-3433		Antonio	Zamora	473-7110	
3				Bernardo	Jaramillo	652-8968	
4				Diana	Cardona	982-3433	
5				Enrique	Montes	901-9724	
6				Celia	Toledo	288-8167	
7							

En este ejemplo estamos buscando el teléfono de “Diana” y ya sabemos, por la función COINCIDIR, que se encuentra en la posición 3 bajo la columna Nombre. Ahora utilizamos la función INDICE sobre los datos de la columna Teléfono y en su segundo argumento le indicamos que deseamos obtener la posición devuelta por la función COINCIDIR. El resultado es correcto y la fórmula recién creada es útil para obtener el teléfono de cualquier otra persona que indiquemos en la celda B1:

B2      fx      =INDICE(F2:F6,COINCIDIR(B1,D2:D6,0))							
	A	B	C	D	E	F	G
1	Nombre:	Antonio		Nombre	Apellido	Teléfono	
2	Teléfono:	473-7110		Antonio	Zamora	473-7110	
3				Bernardo	Jaramillo	652-8968	
4				Diana	Cardona	982-3433	
5				Enrique	Montes	901-9724	
6				Celia	Toledo	288-8167	
7							



A diferencia de la función BUSCARV, la fórmula con INDICE y COINCIDIR no nos obliga a realizar la búsqueda en la primera columna de un rango y para demostrarlo resolveremos el caso donde deseamos encontrar el nombre de una persona a partir de su apellido.

B2		=INDICE(D2:D6,COINCIDIR(B1,E2:E6,0))					
	A	B	C	D	E	F	G
1	Apellido:	Montes		Nombre	Apellido	Teléfono	
2	Nombre:	Enrique		Antonio	Zamora	473-7110	
3				Bernardo	Jaramillo	652-8968	
4				Diana	Cardona	982-3433	
5				Enrique	Montes	901-9724	
6				Celia	Toledo	288-8167	
7							

La diferencia entre esta fórmula y la del ejemplo anterior son los rangos de búsqueda. La función COINCIDIR hace la búsqueda sobre la columna Apellido y devuelve la posición correcta para que la función INDICE devuelva la misma posición pero bajo la columna Nombre. Este tipo de búsqueda no es posible de realizar con la función BUSCARV porque ya hemos visto que dicha función siempre hace la búsqueda sobre la primera columna del rango.

La flexibilidad que nos ofrece la función INDICE nos permite utilizarla en otros escenarios para [buscar un valor y regresar múltiples resultados](#). Por otro lado, la función COINCIDIR combinada con la función BUSCARV es una excelente alternativa cuando deseamos hacer una búsqueda sobre una tabla y elegir la columna que necesitamos como resultado a partir de una lista desplegable. Si quieres leer más sobre esta alternativa consulta el artículo [Combinar BUSCARV y COINCIDIR](#).

Ahora ya conoces dos alternativas para crear una fórmula para buscar datos en Excel, ya sea utilizando la función BUSCARV o la combinación de las funciones INDICE y COINCIDIR. Experimenta con ambas opciones y elige la que mejor se adapte a tus necesidades de acuerdo a los beneficios que ofrece cada una de ellas.


## 2. BUSCAR VALORES USANDO FORMATO CONDICIONAL

Seguramente utilizas las hojas de Excel para almacenar grandes cantidades de datos, como el catálogo de clientes de la empresa, el registro de las ventas de productos, etc. Con esa gran cantidad de información es indispensable encontrar los datos precisos en todo momento.

En esta ocasión te mostraré cómo utilizar el formato condicional para buscar dentro de una hoja de Excel y resaltar las celdas coincidentes. Observa la siguiente imagen:


	A	B	C	D
1				
2				
3				
4		Adalberto Hernández		
5		Adriana Oropeza		
6		Andrés Reynoso		
7		Benjamín Arias		
8		Berenice Ramos		
9		Camila Calderón		
10		Carlos Villarreal		
11		Daniel Rentería		
12		David González		
13		Diana García		
14		Eduardo Navarrete		
15		Elena Rocha		
16		Eliseo Fernández		
17		Erick Rodríguez		
18		Fabiola Álvarez		
19		Fernanda Jiménez		
20		Isabel Luna		
21		Jimena Palacios		
22		Juan Pedraza		
23		Katia Zamora		
24				

Para este ejemplo utilizaré los datos de la columna B y utilizaré la celda B2 como el cuadro de búsqueda, es decir, la celda donde colocaré el término/texto que deseo encontrar en los datos. Ahora debemos crear la regla que aplique el formato condicional y para eso se deben seleccionar las celdas con los datos e ir al comando Formato condicional y seleccionar la opción Nueva regla:



Se mostrará el cuadro de diálogo *Nueva regla de formato* donde deberás seleccionar la opción *Utilice una fórmula que determine las celdas para aplicar formato*. En la caja de texto coloca la siguiente fórmula, que explicaré en un instante:

```
=Y ($B$2<>"", ENCONTRAR (MINUSC ($B$2) , MINUSC (B4) ) )
```



También haz clic en el botón Formato... para seleccionar el formato que se aplicará a las celdas que cumplan la regla.

La fórmula que acabas de escribir compara el texto de la celda de búsqueda (\$B\$2) con las celdas que contienen los datos. La comparación se hace a través de la [función ENCONTRAR](#) que compara ambos valores. Puedes observar también que para ambos valores utilicé la función MINUSC que los convierte en minúsculas antes de hacer la comparación de manera que la búsqueda no sea sensible a mayúsculas y minúsculas.

Finalmente, la intención de la función Y es evitar que el formato se aplique a todas las celdas cuando la celda \$B\$2 esté vacía.

Ahora haz clic en el botón Aceptar y podrás comenzar a encontrar los valores que introduzcas en la celda de búsqueda.

	A	B	C
1			
2		Diana +	
3			
4		Adalberto Hernández	
5		Adriana Oropeza	
6		Andrés Reynoso	
7		Benjamín Arias	
8		Berenice Ramos	
9		Camila Calderón	
10		Carlos Villarreal	
11		Daniel Rentería	
12		David González	
13		Diana García	
14		Eduardo Navarrete	
15		Elena Rocha	
16		Eliseo Fernández	
17		Erick Rodríguez	
18		Fabiola Álvarez	
19		Fernanda Jiménez	
20		Isabel Luna	
21		Jimena Palacios	
22		Juan Pedraza	
23		Katia Zamora	
24			

### 3. CÓMO COMPARAR DOS LISTAS EN EXCEL

Una tarea muy común es comparar dos listas en Excel para conocer las semejanzas o las diferencias entre ambas listas. Aunque existen varias maneras de realizar una comparación entre listas, en esta ocasión utilizaré el Formato condicional.

Vamos a suponer las siguientes dos listas en Excel:


	A	B
1	<b>Lista 1</b>	<b>Lista 2</b>
2	Abel	Caro
3	Caro	Dany
4	Hugo	Gaby
5	Irma	Hugo
6	Juan	Irma
7	Luis	Javi
8	Mara	Juan
9	Paco	Lula
10	Paty	Nery
11	Vica	Paty
12		

El primer ejemplo será conocer los elementos de la Lista 2 que están presentes en la Lista 1 y para ello debemos seguir los siguientes pasos:

1. Seleccionar el rango de datos B2:B11.
2. En la ficha Inicio pulsamos el botón Formato Condicional y seleccionamos la opción Nueva regla.
3. Dentro de la lista de tipos de regla seleccionamos la opción Utilice una fórmula que determine las celdas para aplicar formato.
4. En el cuadro de fórmula colocamos lo siguiente:

```
=COINCIDIR(B2, $A$2:$A$11, 0) > 0
```

5. Hacemos clic en el botón Formato y seleccionamos un color de relleno.



Al momento de aplicar esta regla de formato condicional obtenemos el siguiente resultado:


	A	B	C
1	<b>Lista 1</b>	<b>Lista 2</b>	
2	Abel	Caro	
3	Caro	Dany	
4	Hugo	Gaby	
5	Irma	Hugo	
6	Juan	Irma	
7	Luis	Javi	
8	Mara	Juan	
9	Paco	Lula	
10	Paty	Nery	
11	Vica	Paty	
12			

Puedes observar que las celdas de la Lista 2 que tienen el fondo de color verde son precisamente aquellos valores que también están presentes en la Lista 1.

Es evidente que las celdas que no tienen color de fondo son aquellas que no forman parte de la Lista1 así que vamos a hacer el ejemplo de ese caso en particular. Nuestro segundo ejemplo es conocer los elementos de la Lista2 que no están en la Lista 1.

En este segundo ejemplo seguimos los mismos pasos que en el ejemplo anterior pero utilizamos la siguiente fórmula:

```
=ESERROR(COINCIDIR(B2, $A$2:$A$11, 0))
```





Observa el resultado de aplicar esta regla de formato condicional:

	A	B	C
1	<b>Lista 1</b>	<b>Lista 2</b>	
2	Abel	Caro	
3	Caro	Dany	
4	Hugo	Gaby	
5	Irma	Hugo	
6	Juan	Irma	
7	Luis	Javi	
8	Mara	Juan	
9	Paco	Lula	
10	Paty	Nery	
11	Vica	Paty	
12			

En este segundo ejemplo, las celdas que tienen el fondo de color son aquellos valores de la Lista 2 que no están presentes en la Lista 1. Con ambos ejemplos hemos analizado los dos casos posibles: encontrar los valores de la Lista 2 que son similares a los de la Lista 1 y también encontrar aquellos que son diferentes.

Si quieres hacer las comparaciones tomando como referencia la Lista1 los pasos serán los mismos que en los ejemplos anteriores a excepción de que iniciaremos seleccionando el rango de la Lista 1 y que debemos cambiar las fórmulas utilizadas en la regla de formato condicional de la siguiente manera.

Para encontrar valores de la Lista1 que están presentes en la Lista2:

```
=COINCIDIR(A2, $B$2:$B$11, 0) > 0
```

Para encontrar valores de la Lista1 que no están en la Lista2:

```
=ESERROR(COINCIDIR(A2, $B$2:$B$11, 0))
```

	A	B
1	<b>Lista 1</b>	<b>Lista 2</b>
2	Abel	Caro
3	Caro	Dany
4	Hugo	Gaby
5	Irma	Hugo
6	Juan	Irma
7	Luis	Javi
8	Mara	Juan
9	Paco	Lula
10	Paty	Nery
11	Vica	Paty
12		

```
=COINCIDIR(A2, $B$2:$B$11, 0) > 0
```

	A	B
1	<b>Lista 1</b>	<b>Lista 2</b>
2	Abel	Caro
3	Caro	Dany
4	Hugo	Gaby
5	Irma	Hugo
6	Juan	Irma
7	Luis	Javi
8	Mara	Juan
9	Paco	Lula
10	Paty	Nery
11	Vica	Paty
12		

```
=ESERROR(COINCIDIR(A2, $B$2:$B$11, 0))
```

Otro método que puedes utilizar para comparar dos listas en Excel es a través de la consolidación de datos. La ventaja de ese método es que podrás comparar tres o más listas. Lee todo el detalle sobre dicho método en el artículo [Comparar listas consolidando datos](#).

## 4. FÓRMULA PARA OBTENER EL ÚLTIMO DÍA DEL MES

Frecuentemente necesitamos calcular el último día del mes en Excel, ya sea para calcular los días faltantes o para tomar esa fecha como base para algún cálculo. En esta ocasión mostraré dos métodos para obtener dicha fecha.

El primer método es utilizando la función FECHA la cual permite obtener el número de serie de una fecha determinada. Supongamos que tenemos la fecha 19 de Julio de 2011 y queremos saber el último día del mes.

Podemos utilizar la función de la siguiente manera:

```
=FECHA(2011, 7 + 1, 0)
```

Pon especial atención al segundo parámetro de la función que indica el mes. Julio es el mes 7 y es por eso que observas el número 7, pero estoy sumando un mes adicional lo cual daría el mes 8 (Agosto). La razón de sumar 1 es porque en el tercer parámetro colocaré el valor 0 (cero) y lo explico a continuación.

Al especificar el día cero del mes ocho estaré pidiendo a Excel la fecha 0 de Agosto del 2011, lo cual identificará de inmediato como una fecha inválida, pero en lugar de indicar un error Excel interpreta el día cero como el día anterior al 1 de Agosto del 2011 y por lo tanto me dará como resultado el último día del mes de Julio que es el mes anterior.

<i>fx</i>	=FECHA(2011, 7 + 1, 0)	
	C	D
	31/07/2011	

El segundo método es mucho más directo que el anterior ya que la función FIN.MES ha sido diseñada específicamente para este fin. Para obtener el último día del mes de una fecha especificada podemos utilizar la función FIN.MES de la siguiente manera:

```
=FIN.MES (FECHA (2011,7,19) , 0)
```

El primer parámetro de la función FIN.MES es una fecha por lo que vuelvo a utilizar la función FECHA para obtener el valor adecuado a partir del número de año, mes y día. El segundo parámetro es el número de meses que deseas añadir a la fecha especificada. El resultado es el siguiente.

<i>f<sub>x</sub></i>	=FIN.MES(FECHA(2011,7,19),0)		
	C	D	E
	31/07/2011		
	40755		

Recuerda que la función devuelve el número de serie de la fecha y por lo tanto Excel no aplicará el formato que esperamos ver así que necesitarás dar explícitamente el formato de fecha a la celda.


<i>f<sub>x</sub></i>	=FIN.MES(FECHA(2011,7,19),0)		
	C	D	E
	31/07/2011		
	31/07/2011		

En ocasiones estamos trabajando con días hábiles y necesitamos encontrar el primero o último día hábil del mes. Dicha tarea requiere de métodos diferentes a los presentados anteriormente por lo que te conviene leer el artículo [Primer y último día hábil del mes con Excel](#).

## 5. FÓRMULA PARA OBTENER EL DÍA DE LA SEMANA

Cuando trabajamos con [fechas en Excel](#) frecuentemente necesitamos obtener el día de la semana que le corresponde a una fecha determinada. Por ejemplo, saber el día de la semana que corresponde a la fecha 11/11/2013.


La opción más sencilla para ver el día de la semana de una fecha determinada es modificando el formato de la celda. Por ejemplo, en la siguiente imagen puedes notar que la columna A tiene una serie de fechas en formato dd/mm/aaaa.



He copiado las mismas fechas a la columna B y he modificado el formato de las celdas para mostrar el día de la semana.

Al aplicar el formato de Fecha larga obtenemos el día de la semana para cada fecha. Este formato lo he seleccionado de la lista desplegable que se encuentra en el grupo Número de la ficha Inicio.

Por el contrario, si solo deseamos mostrar el día de la semana, sin mostrar la fecha completa, entonces necesitamos crear un [formato personalizado de celdas](#) que utilice el formato dddd:



De esta manera comprobamos que al modificar el formato de una celda podemos visualizar el día de la semana para una fecha determinada, sin embargo el valor de la celda sigue siendo un valor numérico y eso impide que podamos concatenarlo con una cadena de texto.

Por ejemplo, si en la columna C deseo formar la cadena de texto “Hoy es lunes” utilizando el resultado de la columna B obtendré un resultado inesperado:

C2			<b>f<sub>x</sub></b>	=CONCATENAR("Hoy es ", B2)	
	A	B	C	D	E
1	<b>Fechas</b>				
2	01/11/2013	viernes	Hoy es 41579		
3	02/11/2013	sábado	Hoy es 41580		
4	03/11/2013	domingo	Hoy es 41581		
5	04/11/2013	lunes	Hoy es 41582		
6	05/11/2013	martes	Hoy es 41583		
7	06/11/2013	miércoles	Hoy es 41584		
8	07/11/2013	jueves	Hoy es 41585		
9	08/11/2013	viernes	Hoy es 41586		
10	09/11/2013	sábado	Hoy es 41587		
11	10/11/2013	domingo	Hoy es 41588		
12	11/11/2013	lunes	Hoy es 41589		
13	12/11/2013	martes	Hoy es 41590		
14	13/11/2013	miércoles	Hoy es 41591		
15	14/11/2013	jueves	Hoy es 41592		
16	15/11/2013	viernes	Hoy es 41593		
17					

Como puedes observar, Excel concatena el valor numérico de la fecha ya que el formato personalizado de la columna B no modificó el valor de la celda sino solamente la manera en que es desplegado en pantalla. Sin embargo, podemos solucionar este problema si utilizamos la función **TEXTO**.

La [función TEXTO](#) nos permite convertir un valor numérico en texto y asignar un formato al momento de la conversión. Observa el resultado al utilizar la función **TEXTO** para convertir las fechas de la columna A utilizando el formato dddd:

B2		<b>f<sub>x</sub></b> =TEXTO(A2,"dddd")		
	A	B	C	D
1	<b>Fechas</b>			
2	01/11/2013	viernes		
3	02/11/2013	sábado		
4	03/11/2013	domingo		

La fórmula nos devuelve el día de la semana de la fecha correspondiente, y en esta ocasión no se modifica el formato de las celdas, sino que la función TEXTO nos devuelve una cadena que podremos concatenar fácilmente.

En la siguiente imagen utilizo el resultado de la función TEXTO para concatenar el día de la semana con la cadena de texto “Hoy es “:

B2		<b>f<sub>x</sub></b>	=CONCATENAR("Hoy es ", TEXTO(A2,"dddd"))		
	A	B	C	D	E
1	<b>Fechas</b>				
2	01/11/2013	Hoy es viernes			
3	02/11/2013	Hoy es sábado			
4	03/11/2013	Hoy es domingo			
5	04/11/2013	Hoy es lunes			
6	05/11/2013	Hoy es martes			
7	06/11/2013	Hoy es miércoles			
8	07/11/2013	Hoy es jueves			
9	08/11/2013	Hoy es viernes			
10	09/11/2013	Hoy es sábado			
11	10/11/2013	Hoy es domingo			
12	11/11/2013	Hoy es lunes			
13	12/11/2013	Hoy es martes			
14	13/11/2013	Hoy es miércoles			
15	14/11/2013	Hoy es jueves			
16	15/11/2013	Hoy es viernes			
17					

La función TEXTO no solo nos ayuda a obtener el día de la semana de una fecha sino que también podemos obtener el nombre del mes y el año. Si quieres saber más sobre el uso de la función TEXTO junto con las fechas en Excel te recomiendo leer el artículo [Extraer información de una fecha](#).

En los ejemplos anteriores hemos visto cómo obtener el nombre del día de la semana para una fecha determinada, pero si para ti es suficiente con el número del día de la semana, entonces puedes utilizar la función DIASEM.



De manera predeterminada la [función DIASEM](#) asigna el número 1 al día domingo, hasta llegar al día sábado que tendrá asignado el número 7. En la siguiente imagen puedes observar el resultado de utilizar esta función con las fechas de la columna A:

C4		<b>f<sub>x</sub></b>	=DIASEM(A4)	
	A	B	C	D
1	<b>Fechas</b>			
2	01/11/2013	Hoy es viernes	6	
3	02/11/2013	Hoy es sábado	7	
4	03/11/2013	Hoy es domingo	1	
5	04/11/2013	Hoy es lunes	2	
6	05/11/2013	Hoy es martes	3	
7	06/11/2013	Hoy es miércoles	4	
8	07/11/2013	Hoy es jueves	5	
9	08/11/2013	Hoy es viernes	6	
10	09/11/2013	Hoy es sábado	7	
11	10/11/2013	Hoy es domingo	1	
12	11/11/2013	Hoy es lunes	2	
13	12/11/2013	Hoy es martes	3	
14	13/11/2013	Hoy es miércoles	4	
15	14/11/2013	Hoy es jueves	5	
16	15/11/2013	Hoy es viernes	6	
17				

Si por el contrario deseas considerar el inicio de semana en lunes, solo debes utilizar el valor 2 como el segundo argumento de la función DIASEM. Una ventaja de utilizar los números de los días, en lugar de sus nombres, es que podemos hacer comparaciones fácilmente y tomar decisiones al respecto.

Por ejemplo, si deseo mostrar el mensaje “Descanso” cuando la fecha sea un sábado o un domingo, puedo utilizar la siguiente fórmula:

C3	=SI(DIASEM(A3,2)>5,"Descanso", "Día laboral")				
	A	B	C	D	E
1	Fechas				
2	01/11/2013	Hoy es viernes	Día laboral		
3	02/11/2013	Hoy es sábado	Descanso		
4	03/11/2013	Hoy es domingo	Descanso		
5	04/11/2013	Hoy es lunes	Día laboral		
6	05/11/2013	Hoy es martes	Día laboral		
7	06/11/2013	Hoy es miércoles	Día laboral		
8	07/11/2013	Hoy es jueves	Día laboral		
9	08/11/2013	Hoy es viernes	Día laboral		
10	09/11/2013	Hoy es sábado	Descanso		
11	10/11/2013	Hoy es domingo	Descanso		
12	11/11/2013	Hoy es lunes	Día laboral		
13	12/11/2013	Hoy es martes	Día laboral		
14	13/11/2013	Hoy es miércoles	Día laboral		
15	14/11/2013	Hoy es jueves	Día laboral		
16	15/11/2013	Hoy es viernes	Día laboral		
17					

En esta fórmula utilizo el valor 2 para el segundo argumento de la función DIASEM y por lo tanto obtendré el valor 6 y 7 para sábado y domingo respectivamente. De esta manera la [función SI](#) mostrará el mensaje “Descanso” para dichos días o de lo contrario mostrará el mensaje “Día laboral”.

Con los ejemplos anteriores he mostrado cómo obtener el día de la semana en Excel según la fecha indicada. Utiliza las fórmulas desarrolladas para obtener el nombre del día de la semana o simplemente para trabajar con el número del día.

## 6. FÓRMULA PARA CALCULAR LA EDAD

Calcular la edad con Excel puede llegar a ser un poco complicado porque los cálculos no solo dependen del año sino también del día actual. La solución se complica cuando te das cuenta que es necesario considerar los años bisiestos.

En esta ocasión presentaré tres maneras diferentes de calcular la edad con Excel. Todas las fórmulas asumen que la celda A2 contiene la fecha nacimiento y que la celda B2 tiene la fecha actual que es calculada por la función HOY.

B2		$f_x$	=HOY()	
	A	B	C	D
1	Fecha Nac.	Hoy	Edad	
2	20/04/1984	07/02/2012		
3				

El primer método obtiene la diferencia en días entre ambas fechas y el resultado lo divide entre 365.25. Este último número es porque cada cuatro años tenemos un año con 366 días (bisiesto) por lo que 365.25 es el promedio de los cuatro años. La función ENTERO eliminará los decimales del resultado.

C2		$f_x$	=ENTERO((B2-A2)/365.25)		
	A	B	C	D	E
1	Fecha Nac.	Hoy	Edad		
2	20/04/1984	07/02/2012	27		
3					

La desventaja de este método es que no es muy exacto y tiene problemas serios cuando intentas calcular la edad de niños.

Por ejemplo, para un niño que tenga exactamente 1 año de edad cumplido obtendremos el resultado 0 (cero) por haber realizado la división por 365.25.

La función FRAC.AÑO devuelve la fracción de un año a partir del número total de días que existen entre dos fechas. Observa el resultado.

C3		fx		=ENTERO(FRAC.AÑO(A3,B3))		
	A	B	C	D	E	F
1	Fecha Nac.	Hoy	Edad			
2	20/04/1984	07/02/2012	27			
3	20/04/1984	07/02/2012	27			
4						

De igual manera he utilizado la función ENTERO para remover los decimales del resultado.

La función SIFECHA es el método más exacto para calcular la edad con Excel. Para obtener los años de diferencia entre dos fechas utilizamos el argumento "y" en la función:

C4		fx		=SIFECHA(A4,B4,"y")		
	A	B	C	D	E	
1	Fecha Nac.	Hoy	Edad			
2	20/04/1984	07/02/2012	27			
3	20/04/1984	07/02/2012	27			
4	20/04/1984	07/02/2012	27			
5						

Así que solamente elige el método que te parezca más adecuado para calcular la edad con Excel e impleméntalo.

## 7. LA FUNCIÓN SIFECHA EN EXCEL

La función SIFECHA es una función en Excel a la cual no tenemos acceso desde el catálogo de funciones y sin embargo ha estado presente en las últimas versiones de Excel.

Esta función nos ayuda a obtener la diferencia entre dos fechas y además nos permitirá elegir el tipo de dato que deseamos como resultado y que pueden ser días, meses o años. La sintaxis de la función SIFECHA es la siguiente:

```
=SIFECHA(fecha1, fecha2, tipo)
```

El primero y segundo argumentos son fechas en donde la fecha1 tiene que ser menor que fecha2. Dicho de otra manera, la fecha2 debe ser la fecha más reciente. Si estos parámetros se especifican al revés, entonces obtendremos un error del tipo #¡NUM!

El tercer argumento de la función SIFECHA es de suma importancia porque es donde especificaremos la unidad de medición de tiempo que deseamos utilizar. A continuación la tabla de posibles valores para este tercer argumento:

Valor	Significado	Descripción
"d"	Días	Número de días entre la fecha1 y la fecha2.
"m"	Meses	Número de meses entre la fecha1 y la fecha2.
"y"	Años	Número de años entre la fecha1 y la fecha2.
"yd"	Días excluyendo años	Número de días entre la fecha1 y la fecha2 donde ambas fechas tienen el mismo año.

"ym"	Meses excluyendo años	Número de meses entre la fecha1 y la fecha2 donde ambas fechas tienen el mismo año.
"md"	Días excluyendo meses y años	Número de días entre la fecha1 y la fecha2 donde ambas fechas tienen el mismo mes y el mismo año.

A continuación muestro un ejemplo de uso de la función cuando tenemos ambas fechas en diferentes años:

B5		$f_x$	=SIFECHA(A2,B2,A5)	
	A	B	C	D
1	Inicio	Fin		
2	15/08/2005	26/04/2011		
3				
4	Argumento	Resultado	Fórmula	
5	d	2080	=SIFECHA(A2,B2,A5)	
6	m	68	=SIFECHA(A2,B2,A6)	
7	y	5	=SIFECHA(A2,B2,A7)	
8				

Si las fechas que vas a comparar tienen el mismo año, puedes utilizar los argumentos yd y ym:

B5		$f_x$	=SIFECHA(A2,B2,A5)	
	A	B	C	D
1	Inicio	Fin		
2	15/08/2010	26/11/2010		
3				
4	Argumento	Resultado	Fórmula	
5	yd	103	=SIFECHA(A2,B2,A5)	
6	ym	3	=SIFECHA(A2,B2,A6)	
7				
8				

El último argumento md es utilizado en caso de que tengas dos fechas tanto del mismo mes como del mismo año.

Aunque existen varias alternativas de argumentos para la función, recomiendo utilizar los argumentos [d, m, y] porque son los que funcionan correctamente sin importar el año o mes de las fechas.

Por último debo mencionar que si estás utilizando la versión en inglés de Excel, encontrarás esta función con el nombre DATEDIF y su funcionamiento es exactamente igual al mostrado en este artículo.

Recuerda que no encontrarás a la función SIFECHA dentro del listado de funciones de Excel y tampoco se mostrará ayuda al ingresarla en la barra de fórmulas. Solo debes confiar en que dicha función está disponible en Excel y comenzar a utilizarla en tus fórmulas.

El uso de la función SIFECHA es de gran utilidad especialmente para obtener las diferencias en meses o años entre dos fechas ya que dicho cálculo requiere de mucha pericia para ejecutarlo correctamente debido a las diferencias en el número de días que tiene cada mes del año.

## 8. SUMAR RANGOS QUE CONTIENEN ERRORES

Si tienes una lista de número que deseas sumar pero por alguna razón una de las celdas contiene un error, la función SUMA no funcionará correctamente y te devolverá un error #N/A. En esta ocasión te mostraré cómo sumar rangos que contienen errores en Excel.

En la siguiente imagen puedes observar cómo la función SUMA devuelve un error al encontrar errores en alguna de las celdas del rango que está intentando sumar.

D1      fx    =SUMA(A1:A10)				
	A	B	C	D
1	1		Suma de A1:A10	#N/A
2	2			
3	3			
4	4			
5	#N/A			
6	6			
7	7			
8	#¡DIV/0!			
9	9			
10	10			
11				

Es suficiente con que exista una sola celda con error dentro del rango para que la función SUMA genere un error. Existen dos alternativas para solucionar este problema.

Antes de mostrar la primera alternativa de solución a nuestro problema tengo que demostrar que la función SUMA no tiene problema alguno con las celdas vacías. Observa lo que sucede si en lugar de tener errores tenemos celdas vacías:



D1      fx      =SUMA(A1:A10)				
	A	B	C	D
1	1		Suma de A1:A10	42
2	2			
3	3			
4	4			
5				
6	6			
7	7			
8				
9	9			
10	10			
11				

En este caso la función SUMA devuelve el resultado correcto de la suma para el rango A1:A10. Una vez demostrado esto plantearé la solución para sumar rangos que contienen errores en Excel utilizando la función SI.ERROR.

La función SI.ERROR nos permite evaluar si el valor de una celda contiene un error en cuyo caso nos permitirá indicar el valor de retorno. Para resolver nuestro problema será suficiente pedir a la función SI.ERROR que devuelva un valor vacío al momento de encontrar un error. La fórmula que utilizaremos será la siguiente:

```
=SUMA(SI.ERROR(A1:A10, ""))
```

Ya que deseamos aplicar esta fórmula al rango A1:A10 necesitamos utilizarla como una [fórmula matricial](#) por lo que deberás pulsar la combinación de teclas Ctrl + Shift + Entrar después de haber ingresado la fórmula. El resultado es el siguiente:

D1 <span>fx</span> {=SUMA(SI.ERROR(A1:A10, ""))}					
	A	B	C	D	E
1	1		Suma de A1:A10	42	
2	2				
3	3				
4	4				
5	#N/A				
6	6				
7	7				
8	#iDIV/0!				
9	9				
10	10				
11					

Podemos utilizar esta misma técnica en caso de querer obtener el promedio del rango. Solo debemos utilizar la función PROMEDIO en lugar de la función SUMA:

```
=PROMEDIO (SI.ERROR (A1:A10, ""))
```

La otra alternativa para sumar rangos que contienen errores es utilizar la función AGREGAR.

La función AGREGAR está disponible a partir de Excel 2010 y simplifica mucho la solución de problemas como el de la suma de rangos con errores. Para nuestro ejemplo podemos utilizar la siguiente fórmula:

```
=AGREGAR (9, 6, A1:A10)
```

El primer argumento de la función indica la operación a realizar (9=SUMA). El segundo argumento nos permite configurar algunas opciones de la función y en este caso el número 6 le indica a la función que omita los valores de error.

El último argumento es la referencia al rango sobre el cual se efectuará la operación. Observa el resultado de esta fórmula:

D1      fx      =AGREGAR(9,6,A1:A10)				
	A	B	C	D
1	1		Suma de A1:A10	42
2	2			
3	3			
4	4			
5	#N/A			
6	6			
7	7			
8	#¡DIV/0!			
9	9			
10	10			
11				

También podemos obtener el promedio de este rango con errores utilizando la función AGREGAR. Solo debemos cambiar el primer argumento por el valor 1 que indica a la función que aplique la operación PROMEDIO sobre el rango.

```
=AGREGAR(1,6,A1:A10)
```

Para más información sobre esta función consulta el artículo [La función AGREGAR en Excel](#).

## 9. FÓRMULA PARA CONTAR PALABRAS EN EXCEL

Si estás en la necesidad de contar palabras en Excel este artículo puede ser de ayuda ya que en esta ocasión revisaremos un método para contar las palabras que existen dentro de una celda. Supongamos que en la celda A1 tienes un texto y deseas saber el número de palabras que contiene dicha celda.

	A	B
1	Este es el texto que contiene un enunciado muy largo y donde se encuentran todas las palabras que necesito contar.	
2		

Para alcanzar nuestro objetivo utilizaremos 3 funciones: ESPACIOS, SUSTITUIR y LARGO.

La [función ESPACIOS](#) es de mucha ayuda en esta tarea porque removerá todos los espacios dobles que puedan existir entre las palabras. Además removerá cualquier espacio al inicio o final del enunciado.

Esta fórmula la utilizaré cada vez que necesite referirme al contenido de la celda A1 de la siguiente manera:

```
=ESPACIOS (A1)
```

Una vez aplicada la función ESPACIOS tendré la seguridad de que absolutamente todas las palabras están separadas por solo un espacio en blanco lo cual hará mi tarea más sencilla.

Ahora ya conocemos los beneficios de la función ESPACIOS, la utilizaré junto con la [función LARGO](#) para contar el número de caracteres del enunciado de la celda A1:

C2		fx	=LARGO(ESPACIOS(A1))	
	A	B	C	D
1	Este es el texto que contiene un enunciado muy largo y donde se encuentran todas las palabras que necesito contar.			
2		Número de caracteres:	114	
3				
4				

En un segundo cálculo utilizaré la función SUSTITUIR para remover los espacios en blanco y volveré a contar el número de caracteres.

C3		fx		=LARGO(SUSTITUIR(ESPACIOS(A1), " ", ""))	
	A	B	C	D	
1	Este es el texto que contiene un enunciado muy largo y donde se encuentran todas las palabras que necesito contar.				
2		Número de caracteres:	114		
3		Número de caracteres (sin espacios):	95		
4					

Observa cómo vuelvo a utilizar la función ESPACIOS para garantizar que solo exista un solo espacio entre cada palabra y después remuevo todos los espacios con la función SUSTITUIR. El número total de caracteres me lo da de nuevo la función LARGO.

Hemos llegado prácticamente al final del cálculo porque contar palabras en Excel es casi lo mismo que contar los espacios que existen dentro de un enunciado.

Esta deducción es muy simple ya que cada palabra está siempre precedida por un espacio, por lo que al contar los espacios estaré muy cerca de conocer el número total de palabras.

La única palabra que no está precedida por un espacio es la palabra inicial y es por eso que el número de palabras de un enunciado será su número de espacios más uno.

En nuestro ejemplo, la resta de las fórmulas que hemos calculado hasta ahora me dará el número de espacios dentro del enunciado:

C4      fx      =C2-C3			
	A	B	C
1	Este es el texto que contiene un enunciado muy largo y donde se encuentran todas las palabras que necesito contar.		
2		Número de caracteres:	114
3		Número de caracteres (sin espacios):	95
4		Espacios en el enunciado:	19
5			

El número total de palabras en el enunciado de la celda A1 será el número de espacios más uno.

C5      fx      =C4+1			
	A	B	C
1	Este es el texto que contiene un enunciado muy largo y donde se encuentran todas las palabras que necesito contar.		
2		Número de caracteres:	114
3		Número de caracteres (sin espacios):	95
4		Espacios en el enunciado:	19
5		Número de palabras:	20
6			
7			

Si quieres realizar el mismo cálculo pero en una sola celda, sin necesidad de ocupar varias celdas como en el ejemplo, es suficiente con utilizar la siguiente fórmula que agrupa todos los conceptos revisados anteriormente:

```
=LARGO(ESPACIOS(A1)) - LARGO(SUSTITUIR(ESPACIOS(A1), " ", "")) + 1
```


Solamente sustituye las referencias a la celda A1 por la celda que contiene el enunciado del cual deseas contar el número de palabras.

## 10. LISTA DESPLEGABLE CON DATOS DE OTRA HOJA

Las listas desplegables en Excel son esenciales para facilitar al usuario la elección de valores y mantener la consistencia de nuestros datos. En ocasiones necesitamos crear listas desplegables que tengan como fuente los datos ubicados en otra hoja de Excel.


En este artículo analizaremos, en primer lugar, la creación de una lista desplegable utilizando validación de datos y posteriormente analizaremos la solución utilizando un control de formulario.

Si utilizamos una [lista de validación de datos](#) podemos hacer referencia a los datos de otra hoja al momento de indicar el criterio de validación. En la siguiente imagen puedes notar que estoy creando una lista (en la Hoja2) que hace referencia al rango de A2:A13 de la Hoja1:




Como resultado obtendremos una lista desplegable con los valores contenidos en una hoja diferente:






En el ejemplo anterior creamos una lista desplegable que hacía referencia a un rango de celdas, pero es muy común que nuestros datos estén contenidos en una tabla. Supongamos que los datos de la Hoja1 tienen formato de tabla:




Como sabes, al [crear una tabla en Excel](#) tendrá un nombre asignado que podremos consultar desde el administrador de nombres. De manera predeterminada Excel nombrará las tablas de nuestro libro como Tabla1, Tabla2, Tabla3, etc.

Para crear una lista desplegable con los datos de la columna Meses tenemos que crear un nombre intermedio que haga referencia a la columna que deseamos considerar así que abriré el administrador de nombres (en la ficha Fórmulas) y pulsaré el botón Nuevo.

Observa con detenimiento la referencia que colocaré para el nombre ListaDesplegable:




A la referencia Tabla1[Meses] se le conoce como [referencia estructurada](#) y es la manera en que podemos referirnos a los datos contenidos en una tabla de Excel. Ahora que hemos creado este nombre intermedio podemos crear nuestra lista de validación de datos utilizando como fuente este nuevo nombre:




Al momento de seleccionar el cuadro de texto Origen puedes pulsar la tecla F3 y se mostrará el cuadro de diálogo Pegar nombre o simplemente puedes introducir manualmente el nombre recién creado. Ya que los nombres tienen un alcance en todo el libro podemos crear nuestra lista desplegable con datos de otra hoja sin la necesidad de indicar directamente la referencia al rango que contiene los datos.

Este truco del nombre intermedio es necesario porque Excel no permite colocar la referencia estructurada directamente en el Origen de la lista de validación de datos. Si intentas hacerlo Excel devolverá un Error en la fórmula.

Otra manera de crear una lista desplegable en Excel es utilizar el control de formulario llamado Cuadro combinado. Una vez que agregamos este control a nuestra hoja solo debemos hacer clic derecho sobre él y seleccionar la opción Formato de control y en la sección Control seleccionaremos el Rango que contiene los datos:



Lo único que estoy haciendo es colocar la referencia adecuada hacia la hoja donde se encuentran los datos que deseo incluir. Como resultado obtendremos una lista desplegable con datos de otra hoja:



Recuerda que para conocer la opción que ha sido seleccionada en este tipo de control de formulario debemos vincular el cuadro combinado a una celda la cual reflejará la opción seleccionada. Para conocer un poco más sobre este procedimiento consulta el artículo [Cuadros combinados en Excel](#).

En este artículo hemos analizado tres posibilidades para crear una lista desplegable con datos de otra hoja de manera que no sea necesario tener tanto los datos como las listas desplegables en una misma hoja.

Esta es una práctica muy común, especialmente cuando tenemos múltiples catálogos organizados en diferentes hojas de nuestro libro y necesitamos crear las listas desplegables en una hoja diferente.

## 11. LISTAS DESPLEGABLES DEPENDIENTES

Una de las funcionalidades más utilizadas en la validación de datos en Excel son las listas desplegables ya que nos ofrecen un control absoluto sobre el ingreso de datos de los usuarios. Sin embargo, crear listas dependientes no siempre es una tarea sencilla, así que te mostraré un método para lograr este objetivo.


Decimos que tenemos listas desplegables dependientes cuando la selección de la primera lista afectará las opciones disponibles de la segunda lista. Esto nos ofrece un mayor control sobre las opciones elegidas por el usuario ya que siempre habrá congruencia en los datos ingresados.

Para nuestro ejemplo utilizaremos un listado de países y ciudades con el cual crearemos un par de listas desplegables que mostrarán las ciudades que pertenecen al país previamente seleccionado.

	A	B	C
1	<b>País</b>	<b>Ciudad</b>	
2	Argentina	Buenos Aires	
3	Argentina	Córdoba	
4	Argentina	Rosario	
5	Chile	Santiago	
6	Chile	Concepción	
7	Chile	Valparaíso	
8	Colombia	Bogotá	
9	Colombia	Medellín	
10	Colombia	Cali	
11	España	Madrid	
12	España	Barcelona	
13	España	Valencia	
14	México	Ciudad de México	
15	México	Monterrey	
16	México	Guadalajara	
17	Perú	Lima	
18	Perú	Arequipa	
19	Perú	Trujillo	
20			

Este listado se encuentra en una hoja de Excel llamada Datos que es donde prepararemos los datos de manera que poder crear con facilidad las listas desplegables dependientes desde cualquier otra hoja del libro.

El primer paso que debemos dar es crear una lista de países únicos. Para esto haré una copia de los datos de la columna A y pegaré los valores en la columna D. Posteriormente, con la columna seleccionada, iré a la ficha Datos > Herramientas de datos y pulsaré el botón Quitar duplicados.



Ahora seleccionaré el rango de celdas D2:D7 y le pondré el nombre Países. Para asignar un nombre a un rango de celdas debemos seleccionarlo e ingresar el texto en el Cuadro de nombres de la barra de fórmulas.

Países		f_x		Argentina	
	A	B	C	D	E
1	País	Ciudad		País	
2	Argentina	Buenos Aires		Argentina	
3	Argentina	Córdoba		Chile	
4	Argentina	Rosario		Colombia	
5	Chile	Santiago		España	
6	Chile	Concepción		México	
7	Chile	Valparaíso		Perú	
8	Colombia	Bogotá			
9	Colombia	Medellín			
10	Colombia	Cali			

El segundo paso será nombrar los rangos de las ciudades para cada país de la siguiente manera:

1. Selecciona el rango que contiene las ciudades de un país.
2. Nombra dicho rango con el nombre del país.

Siguiendo este procedimiento tan simple, la siguiente imagen muestra el momento en que selecciono las ciudades de Argentina y asigno el nombre adecuado a dicho rango.


Argentina		f_x		Buenos Aires	
	A	B	C	D	
1	País	Ciudad		País	
2	Argentina	Buenos Aires		Argentina	
3	Argentina	Córdoba		Chile	
4	Argentina	Rosario		Colombia	
5	Chile	Santiago		España	
6	Chile	Concepción		México	
7	Chile	Valparaíso		Perú	
8	Colombia	Bogotá			

Es muy importante que el nombre del rango sea exactamente igual al nombre del país ya que ese será nuestro vínculo entre ambas listas.



De la misma manera como he creado el rango de ciudades para Argentina crearé un nuevo rango para cada país.

Una vez terminada esta tarea tendré 7 rangos nombrados. Un rango nombrado para cada uno de los 6 países y además un nombre para la lista de países únicos. Para ver esa lista de rangos nombrados puedo ir a la ficha Fórmulas y hacer clic en el botón Administrador de nombres.




Si te equivocaste en el nombre del rango o seleccionaste un grupo de celdas incorrecto, el Administrador de nombres te permitirá hacer cualquier modificación haciendo clic en el botón Editar.

Ahora que ya tenemos listos nuestros rangos nombrados podemos crear las listas desplegables. Para eso iré a una nueva hoja de mi libro de Excel,

seleccionaré la celda A2 e iré a la ficha Datos > Herramientas de Datos > Validación de datos.

En el cuadro de diálogo elegiré la opción Lista y en el cuadro Origen colocará el valor “=Países” que es el nombre del rango que contiene la lista de países únicos.



Al hacer clic en el botón Aceptar podremos comprobar que la celda A2 contiene una lista desplegable con los países.


	A	B	C
1	País	Ciudad	
2	<div>▼</div>		
3	Argentina		
4	Chile		
5	Colombia		
6	España		
7	México		
8	Perú		

Ahora crearemos la lista desplegable dependiente de la celda B2 y para eso seleccionaré dicha celda e iré a la ficha Datos > Herramientas de datos > Validación de datos.

En el cuadro de diálogo mostrado seleccionaré la opción *Lista* y en el cuadro *Origen* colocaré la siguiente fórmula:

```
=INDIRECTO (A2)
```

La [función INDIRECTO](#) se encargará de obtener el rango de celdas cuyo nombre coincide con el valor seleccionado en la celda A2.



Es muy probable que al hacer clic en el botón Aceptar se muestre un mensaje de advertencia diciendo que: El origen actualmente evalúa un error ¿Desea continuar?

Este error se debe a que en ese momento no hay un País seleccionado en la celda A2 y por lo tanto la función INDIRECTO devuelve error, así que solo deberás hacer clic en la opción Si para continuar.

En el momento en que selecciones un país de la celda A2, las ciudades de la celda B2 serán modificadas para mostrar solamente aquellas que pertenecen al país seleccionado.

	A	B	C
1	País	Ciudad	
2	Colombia	<input type="text"/>	
3		Bogotá	
4		Medellín	
5		Cali	

Con estos pasos hemos creado un par de listas desplegables dependientes en Excel las cuales muestran las ciudades correspondientes a un país determinado.


Las listas dependientes que acabamos de crear en la sección anterior tienen un pequeño inconveniente y es que después de hacer una primera selección de País y Ciudad, al hacer una nueva selección de País, la celda que muestra las ciudades permanecerá con la selección anterior.

Para que me entiendas mejor hagamos un ejemplo sencillo. Seleccionaré el país Colombia en la celda A2 y posteriormente en la celda B2 seleccionaré la ciudad Medellín. Hasta ahí todo va bien, pero si ahora selecciono el país México en la celda A2, la celda B2 seguirá mostrando la ciudad Medellín.


	A	B	C
1	País	Ciudad	
2	México	Medellín	
3			
4			

Si en ese momento guardamos el libro, tendremos una incongruencia en los datos. La mala noticia es que no existe un comando de Excel para solucionar este problema.

La buena noticia es que podemos utilizar código VBA para pedir a Excel que limpie la celda B2 cada vez que haya un cambio en la celda A2. Para agregar el código debemos hacer clic derecho sobre el nombre de la hoja y seleccionar la opción Ver código.



En las listas desplegables mostradas debemos elegir la opción Worksheet y Change tal como se muestra en la siguiente imagen.



El código que debemos pegar en esta ventana es el siguiente:

```
Private Sub Worksheet_Change(ByVal Target As Range)

If Target = Range("A2") Then
    Range("B2").Value = ""
End If

End Sub
```

El evento Worksheet\_Change se dispara cada vez que se realiza un cambio en una celda de la hoja. Pero ya que estamos interesados en un cambio de la celda A2, comparamos la variable Target para saber si el cambio proviene de dicha celda. En caso afirmativo, limpiamos el valor de la celda B2.


Si aplicas esta solución a tus archivos, deberás guardarlos como un Libro habilitado para macros de manera que pueda ejecutarse adecuadamente el código VBA.

Si deseas agregar nuevos datos a las listas desplegables, deberás tener cuidado de mantener las referencias adecuadas en cada uno de los rangos nombrados. Por ejemplo, para agregar una nueva ciudad para México insertaré una nueva fila debajo de la ciudad Guadalajara.

11	España	Madrid	
12	España	Barcelona	
13	España	Valencia	
14	México	Ciudad de México	
15	México	Monterrey	
16	México	Guadalajara	
17	México	Puebla	
18	Perú	Lima	
19	Perú	Arequipa	



Ahora el país México tiene 4 ciudades en lugar de 3 así que será necesario modificar el rango nombrado para sus ciudades. Para hacer este cambio debemos ir a la ficha Fórmulas y hacer clic en el botón Administrador de nombres. Al abrirse el cuadro de diálogo notarás dos cosas:



1. Aunque las ciudades de Perú fueron desplazadas hacia abajo por la inserción de la nueva fila, Excel modificó automáticamente la referencia para indicar que dicho nombre ahora se refiere el rango B18:B20.
2. Excel no modificó el rango correspondiente a México y en este momento dicho rango termina en la celda B16 por lo que es necesario que modifiquemos manualmente dicha referencia. Para que todo funcione correctamente debo indicar lo siguiente:

```
=Datos!$B$14:$B$17
```

Para ingresar esta nueva referencias puedes seleccionar el nombre México y hacer clic en el botón Editar. Se mostrará un nuevo cuadro de diálogo donde podrás indicar la nueva referencia.

Con este cambio será suficiente para ver la nueva ciudad al momento de seleccionar el país México dentro de las listas desplegables.

	A	B	C
1	País	Ciudad	
2	México	<div>Ciudad de México Monterrey Guadalajara Puebla</div>	
3			
4			
5			
6			


Así que, ya sea que vas a agregar nuevas Ciudades o Países deberás poner especial atención a las referencias de los rangos nombrados y deberás editarlas en caso de ser necesario desde el Administrador de nombres.



## 12. CARGAR VALORES DE LISTA DINÁMICAMENTE

Las listas desplegables nos permiten restringir las opciones que un usuario puede ingresar en una celda, lo cual es de mucha utilidad para validar los datos de nuestra hoja. Sin embargo, las opciones de una lista son determinadas al momento de crearla y en más de una ocasión queremos que esos valores de la lista aumenten o disminuyan automáticamente.


Si aún no sabes cómo crear una lista desplegable en Excel, te recomiendo leer el artículo [Lista para validación de datos en Excel](#) donde aprenderás tres métodos para crear una lista. Uno de esos métodos es ingresar cada uno de los elementos de la lista separados por coma (o por punto y coma) en el recuadro Origen del cuadro de diálogo Validación de datos.



También aprenderás que podemos indicar el origen de la lista como un rango de celdas o como un rango nombrado y Excel obtendrá el valor de cada celda indicada y lo colocará como una opción de la lista.

Lo que nos interesa analizar en esta ocasión es el momento cuando necesitamos actualizar los valores de la lista, ya sea que quieres eliminar un elemento o agregar uno nuevo.

Si has utilizado el método de ingresar los elementos en el recuadro Origen o has indicado un rango de celdas, para poder actualizar los elementos de la lista tendrás que abrir de nuevo el cuadro de diálogo Validación de datos y hacer la modificación manualmente.



La única opción de las que he mencionado anteriormente en donde no será necesario abrir el cuadro de diálogo Validación de datos para actualizar los elementos, serán aquellas listas cuyo origen fue indicado como un rango nombrado.


La gran ventaja de definir el origen de la lista como un rango nombrado, es que podrás tener varias listas en tu libro de Excel basadas en el mismo rango nombrado y eso nos permitirá actualizar todas las listas al mismo tiempo con solo modificar la definición del nombre. Para este primer ejemplo supondremos un rango de datos como el mostrado en la siguiente imagen:

	A1			
	A	B	C	D
1	Adrián	López		
2	Daniel	Rodríguez		
3	Hugo	González		
4	Lucía	Martínez		
5	Mario	Gómez		
6	Martina	García		
7	Pablo	Fernández		
8	Sara	Díaz		
9	Sofía	Pérez		
10	Valeria	Sánchez		
11				

El primer paso para crear nuestra lista es crear el rango nombrado, así que seleccionaré el rango A1:A10 y en el recuadro de nombres colocaré el nombre “Vendedores” tal como se muestra a continuación:

	Vendedores			
	A	B	C	D
1	Adrián	López		
2	Daniel	Rodríguez		
3	Hugo	González		
4	Lucía	Martínez		
5	Mario	Gómez		
6	Martina	García		
7	Pablo	Fernández		
8	Sara	Díaz		
9	Sofía	Pérez		
10	Valeria	Sánchez		
11				

Al pulsar la tecla Entrar habremos creado nuestro rango nombrado y en seguida seleccionaré la celda D1 y pulsaré el comando Datos > Validación de datos donde seleccionaré la opción Lista y colocaré como Origen el nombre recién creado.



Una vez creada la lista, la celda D1 mostrará todos los nombres contenidos en el rango nombrado.

	A	B	C	D	E
1	Adrián	López			
2	Daniel	Rodríguez			
3	Hugo	González			
4	Lucía	Martínez			
5	Mario	Gómez			
6	Martina	García			
7	Pablo	Fernández			
8	Sara	Díaz			
9	Sofía	Pérez			
10	Valeria	Sánchez			
11					

Ahora viene la parte que nos interesa en este artículo y es la actualización de los elementos de la lista. Si quiero agregar un nuevo nombre a la lista debo insertarlo en la fila 11 de la hoja tal como se observa en la siguiente imagen:

	A	B	C
1	Adrián	López	
2	Daniel	Rodríguez	
3	Hugo	González	
4	Lucía	Martínez	
5	Mario	Gómez	
6	Martina	García	
7	Pablo	Fernández	
8	Sara	Díaz	
9	Sofía	Pérez	
10	Valeria	Sánchez	
11	Carla	Ramírez	
12			

Ahora debo ir a la ficha Fórmulas > Nombres definidos > Administrador de nombres y al abrirse el cuadro de diálogo debo pulsar el botón Editar para modificar el nombre “Vendedores” que recién acabamos de crear. Para incluir la nueva fila debemos modificar la referencia del rango nombrado.

Editar nombre

Nombre: Vendedores

Ámbito: Libro

Comentario:

Hace referencia a: =Nombre!\$A\$1:\$A\$11

Aceptar Cancelar

Al hacer clic en Aceptar y cerrar la ventana del Administrador de nombres, la celda D1 mostrará una lista con el nuevo valor que acabamos de incluir en la definición del nombre.

D1        fx					
	A	B	C	D	E
1	Adrián	López			
2	Daniel	Rodríguez		Lucía	
3	Hugo	González		Mario	
4	Lucía	Martínez		Martina	
5	Mario	Gómez		Pablo	
6	Martina	García		Sara	
7	Pablo	Fernández		Sofía	
8	Sara	Díaz		Valeria	
9	Sofía	Pérez		Carla	
10	Valeria	Sánchez			
11	Carla	Ramírez			
12					

Aunque este método no realiza una actualización automática, es de mucha utilidad en caso de tener varias listas basadas en los mismos datos ya que nos evitará el tener que modificar el origen de cada lista y solo tendremos que editar la referencia indicada en el nombre de rango.

El método que te mostraré a continuación hará una actualización automática de los elementos de una lista, ya sea que se agreguen nuevos elementos o que se eliminen. La única condición para que todo funcione correctamente es que se debe utilizar la columna de una hoja exclusivamente para colocar los elementos de la lista.

Se recomienda que esa columna especial no tenga encabezados de columna, ni tampoco celdas con contenido por debajo de los elementos de la lista. Esto se debe a que utilizaremos una fórmula que contará las celdas de dicha columna especial y por lo tanto necesitamos que solamente estén presentes los elementos de la lista.

Todo comienza con el uso de la función CONTARA, la cual nos ayuda a contar las celdas que tienen un valor dentro de un rango. Si queremos saber las celdas de la columna A que tienen un valor, podemos utilizar la siguiente fórmula:

```
=CONTARA (A:A)
```

La referencia A:A le pide a la función que considere todas las celdas de la columna A. Si utilizamos esta fórmula con nuestros datos de ejemplo, obtendremos el siguiente resultado:

D1		fx =CONTARA(A:A)			
	A	B	C	D	E
1	Adrián	López		10	
2	Daniel	Rodríguez			
3	Hugo	González			
4	Lucía	Martínez			
5	Mario	Gómez			
6	Martina	García			
7	Pablo	Fernández			
8	Sara	Díaz			
9	Sofía	Pérez			
10	Valeria	Sánchez			
11					

La fórmula nos devuelve la cantidad exacta de celdas que tienen un valor en la columna A y ese dato lo podremos utilizar para crear una referencia al rango A1:A10. La fórmula que utilizaremos es la siguiente:


```
=DESREF (A1, 0, 0, CONTARA (A:A) , 1)
```

La [función DESREF en Excel](#) nos ayuda a definir un rango de celdas indicando el ancho y alto a partir de una celda inicial.

En este caso, el primer argumento es la celda inicial y el tercero y cuarto argumentos son el alto y ancho del rango que deseamos obtener.

Lo importante de esta fórmula es el alto del rango, que es devuelto por la función CONTARA, ya que eso nos garantizará que siempre obtendremos el rango ocupado por las celdas de la columna A que a su vez serán los elementos de nuestra lista.

En seguida seleccionaré la celda D1 y pulsaré el comando Datos > Validación de datos y crearé una nueva lista indicando como Origen la fórmula que acabamos de crear:



Como resultado obtendremos una lista desplegable que tendrá todos los elementos de la columna A. Y lo mejor es que, al agregar un nuevo elemento en la columna A, será insertado automáticamente en la lista sin necesidad de realizar alguna acción.



	A	B	C	D	E
1	Adrián	López			
2	Daniel	Rodríguez			
3	Hugo	González			
4	Lucía	Martínez			
5	Mario	Gómez			
6	Martina	García			
7	Pablo	Fernández			
8	Sara	Díaz			
9	Sofía	Pérez			
10	Valeria	Sánchez			
11	Carla	Ramírez			
12	Brenda				
13	Enrique				
14					
15					

La tercera técnica que revisaremos, es una combinación de las dos anteriores, donde aprovecharemos lo mejor de ambos métodos.

Por un lado, las listas creadas con rangos nombrados nos permiten actualizar el origen de datos en un solo lugar y como consecuencia todas las listas basadas en ese Nombre se actualizarán automáticamente.

Por otro lado, la fórmula nos da la oportunidad de obtener automáticamente el rango de celdas que contiene los valores dentro de una columna. Los pasos para crear una lista basada en un rango dinámico son los siguientes:

1. Crear un rango nombrado basado en una fórmula.
2. Crear una lista basada en el rango nombrado del paso anterior.

Para el primer paso iré a la ficha Fórmulas > Nombres definidos > Administrador de nombres y una vez que se abra el cuadro de diálogo, pulsaré el botón Nuevo para crear un nuevo nombre que estará basado en la fórmula que se muestra en la siguiente imagen:

Nombre nuevo

Nombre: MiLista

Ámbito: Libro

Comentario:

Hace referencia a: =DESREF(NombreDinamico!\$A\$1,0,0,CONTARA(NombreDinamico!\$A:\$A),1)

Aceptar Cancelar

Observa que el Nombre que he elegido es MiLista y el ámbito es el Libro, lo que quiere decir que el Nombre podrá ser utilizado en cualquier hoja del libro. La fórmula utilizada es la siguiente:


```
=DESREF(NombreDinamico!$A$1,0,0,CONTARA(NombreDinamico!$A:$A),1)
```

Esta fórmula necesita referencias absolutas porque queremos utilizar el nombre en cualquier hoja del libro.

La referencia NombreDinamico!\$A\$1 le dice a Excel que debe ir a la celda A1 de la hoja NombreDinamico sin importar donde se encuentre.

Una vez que hemos creado el nombre de rango basado en una fórmula, podemos crear nuestra lista desplegable que tendrá como origen el rango nombrado que acabamos de crear.

Solo debo ir a la ficha Datos > Validación de datos e ingresar la siguiente información:



De esta manera la lista de validación de datos se actualizará automáticamente al momento de ingresar nuevos valores bajo la columna A. En la siguiente imagen puedes observar que he agregado cinco nuevos nombres y la lista los despliega automáticamente.

	A	B	C	D	E
1	Adrián	López			
2	Daniel	Rodríguez			
3	Hugo	González			
4	Lucía	Martínez			
5	Mario	Gómez			
6	Martina	García			
7	Pablo	Fernández			
8	Sara	Díaz			
9	Sofía	Pérez			
10	Valeria	Sánchez			
11	Eric				
12	Rubén				
13	Laura				
14	Inés				
15	Paola				
16					


Para probar este método, puedes crear listas desplegables en otras hojas del libro y observarás que se actualizan automáticamente al modificar las celdas de la columna A de la hoja NombreDinamico.

Uno de los métodos que más me gusta para agregar valores a una lista de manera automática, es utilizar la columna de una tabla de Excel como el origen de los datos.

Las [tablas de Excel](#) fueron introducidas en Excel 2007 y se refieren a un objeto que tiene un formato especial pero además, tienen una funcionalidad única. La siguiente imagen muestra nuestros datos de ejemplo después de haber creado una tabla con el comando Inicio > Dar formato como tabla.

	A	B	C	D
1	Nombre ▾	Apellido ▾		
2	Adrián	López		
3	Daniel	Rodríguez		
4	Hugo	González		
5	Lucía	Martínez		
6	Mario	Gómez		
7	Martina	García		
8	Pablo	Fernández		
9	Sara	Díaz		
10	Sofía	Pérez		
11	Valeria	Sánchez		
12				

Excel asigna de manera automática un nombre a la tabla y lo puedes confirmar al abrir el Administrador de nombres desde la ficha Fórmulas > Nombres definidos. La siguiente imagen nos muestra que el nombre de la tabla es Tabla1:



Para crear una lista basada en la columna Nombre de la Tabla1 utilizaremos la siguiente fórmula:

```
=INDIRECTO ("Tabla1 [Nombre] ")
```

La [función INDIRECTO](#) nos devuelve una referencia a partir de una cadena de texto y en este caso, la referencia devuelta será a la columna Nombre de la Tabla1.

Al momento de crear nuestra lista desplegable debemos colocar la fórmula de la siguiente manera.

La lista desplegable mostrará todos los valores de la columna Nombre de la tabla.

	D1				
	A	B	C	D	E
1	Nombre	Apellido			
2	Adrián	López		Adrián	
3	Daniel	Rodríguez		Daniel	
4	Hugo	González		Hugo	
5	Lucía	Martínez		Lucía	
6	Mario	Gómez		Mario	
7	Martina	García		Martina	
8	Pablo	Fernández		Pablo	
9	Sara	Díaz		Sara	
10	Sofía	Pérez			
11	Valeria	Sánchez			
12					

Al agregar nuevas filas a la tabla, la lista desplegable mostrará automáticamente los nuevos elementos de la columna Nombre.

Ya que las tablas tienen un alcance en todo el libro, podrás crear este tipo de listas en cualquier hoja del libro y se actualizarán automáticamente al detectar cambios en los datos de la tabla.

Ahora conoces diferentes alternativas en Excel para cargar valores a una lista de manera automática.